

Solving the TTC 2013 Flowgraphs Case with FunnyQT

Dipl.-Inform. Tassilo Horn

horn@uni-koblenz.de

Institute for Software Technology, University Koblenz-Landau, Germany

FunnyQT is a model querying and model transformation library for the functional Lisp-dialect Clojure providing a rich and efficient querying and transformation API.

This paper describes the FunnyQT solution to the TTC 2013 Flowgraphs Transformation Case. It solves all four tasks, and it has won the *best efficiency award* for this case.

1 Introduction

FunnyQT is a new model querying and transformation approach which is implemented as an API for the functional, JVM-based Lisp-dialect Clojure. It provides several sub-APIs for implementing different kinds of queries and transformations. For example, there is a model-to-model transformation API, and there is an in-place transformation API for writing programmed graph transformations. FunnyQT currently supports EMF and JGraLab models, and it can be extended to other modeling frameworks, too.

For solving the tasks of this transformation case¹, FunnyQT's model transformation API and its polymorphic function API have been used for task 1. Both task 2 and task 3 have been tackled algorithmically using FunnyQT's plain querying and model manipulation APIs. Task 4 has been solved by using FunnyQT's querying API and Clojure metaprogramming.

2 Solution Description

Task 1: JaMoPP to StructureGraph. According to the case description [Hor13], the goal of this task is to transform a fine-granular Java syntax graph conforming to the JaMoPP metamodel [HJSW09] into a much simpler structure graph model that only contains statements and expressions that are neither structured nor subdivided any further. However, the original Java code of these statements and expressions should be reflected in the new elements' `txt` attribute. This model-to-text transformation is described in the next paragraph. Thereafter, the model-to-model transformation creating a structure graph from a JaMoPP model is described.

JaMoPP to Text. This model-to-text transformation is implemented using FunnyQT's polymorphic function API. A polymorphic function is a function that is declared once, and then arbitrary many implementations for concrete metamodel types can be added. When a polymorphic function is called, the actual implementation is determined similarly to the typical dispatch in object-oriented programming languages. If there's no implementation provided for the element's type or one of its supertypes, an exception is thrown.

The function `stmt2str` implements the model-to-text transformation required for solving task 1. It is declared as follows.

¹This FunnyQT solution is available at <https://github.com/tsdh/ttc-2013-flowgraphs> and on SHARE (image TTC13::Ubuntu12LTS_TTC13::FunnyQT.vdi)

```
(declare-polyfn stmt2str [elem])
```

`declare-polyfn` declares a new polymorphic function. Its name is `stmt2str`, and it receives exactly one parameter `elem`. Its task is to create a string representation matching the concrete Java syntax for the provided JaMoPP model element.

After the polymorphic function has been declared, implementations for concrete metamodel types can be added using `defpolyfn`. For example, this is the implementation for JaMoPP elements of type `AssignmentExpression`:

```
(defpolyfn stmt2str 'expressions.AssignmentExpression [ae]
  (str (stmt2str (aget ae :child)) " "
    (stmt2str (aget ae :assignmentOperator)) " "
    (stmt2str (aget ae :value))))
```

The `child` of the assignment expression is some variable, the assignment operator is one of `=`, `+=`, `-=`, `*=`, or `/=`, and `value` is an arbitrary expression. These three components are converted to strings using `stmt2str` which are then concatenated.

All in all, the polymorphic `stmt2str` function consists of 22 implementations for various JaMoPP metamodel types accounting to 107 lines of code. The complete model-to-text transformation is printed in Appendix A.

JaMoPP to Structure Graph. The JaMoPP-to-StructureGraph transformation is implemented using FunnyQT's model-to-model transformation API. This transformation also creates `Var` and `Param` objects as requested by task 3.1.

The transformation starts by defining its name and input and output models.

```
(deftransformation java2flowgraph [[in :emf] [out :emf]]
```

There could be arbitrary many input and output models, and they could be of different kinds, e.g., a transformation could receive a JGraLab TGraph and some EMF model, and create an output EMF model. Here, it gets only the JaMoPP EMF input model which is bound to the variable `in`, and one single structure graph output model bound to `out`, which is also an EMF model.

In the body of such a transformation, arbitrary many rules may be defined. The first one is the `method2method` rule shown in the next listing. The `^:top` metadata preceding the rule name specifies that the rule is a top-level rule. Such rules are applied to all matching elements by the transformation itself, whereas non-top-level rules have to be called explicitly from a top-level rule (directly or indirectly).

```
(^:top method2method [m]
  :from 'members.ClassMethod
  :to [fgm 'flowgraph.Method, ex 'flowgraph.Exit]
  (eset! fgm :txt (stmt2str m)) ;; Invoke the model-to-text transformation
  (eset! ex :txt "Exit")
  (eset! fgm :exit ex)
  (eset! fgm :stmts (map stmt2item (aget m :statements))) ;; transform the statements
  (eset! fgm :def (map param2param (aget m :parameters))) ;; transform the parameters
```

It receives a JaMoPP model element `m`. The `:from` clause dictates that `m` must be of type `ClassMethod` in order for the rule to be applicable. The `:to` clause declares the objects to be created. Here, for a given JaMoPP method, a corresponding flowgraph method and its exit object are created. The remainder of the rule is its body containing arbitrary code to set attributes and references. Here, the `txt` attribute of the new method and its exit are set, the former using the polymorphic `stmt2str` function discussed above. The

method's `stmts` reference is set by applying another rule, `stmt2item`, to the statements of the JaMoPP method. Likewise, the method's parameters are transformed by mapping them to the `param2param` rule for setting the method's `def` reference.

A special kind of rules are generalizing rules such as the one shown in the next listing.

```
(stmt2item [stmt]
  :generalizes [local-var-stmt2simple-stmt condition2if block2block
               return2return while-loop2loop break2break continue2continue
               label2label stmt2simple-stmt])
```

This concept is quite similar to mapping disjunction in QVT Operational Mappings. When this rule is called, the rules specified in the `:generalizes` vector are tried one after the other, and the first applicable one is applied, and its result is returned.

The complete `java2flowgraph` model-to-model transformation consists of 15 rules with 93 lines of code in total. It is printed in Appendix B.

Task 2: Control Flow Analysis. The purpose of this task is to create `cfNext` links between `FlowInstr` elements in the flowgraph model created by the model-to-model transformation realizing task 1. Every such flow instruction should be connected to every other flow instruction that may be the next one in the program's control flow. This challenge has been tackled algorithmically using FunnyQT's plain quering and model manipulation APIs.

The algorithm uses a sequence of statements as intermediate representation to work on realizing a pre-order depth-first traversal with look-ahead through the method's statements. In the general case, every flow instruction in that sequence has to be connected with the immediately following flow instruction in the sequence. For various kinds of statements, special rules are needed. For example, when encountering a block in the sequence (which is no flow instruction), the block is replaced with its contents.

Since the next statement in the sequence might not be a flow instruction but some structured statement like a block, an if-statement, or a loop, there's a helper function `cf-peek`. It receives some element and returns either this element if it is a flow instruction, or otherwise the first flow instruction inside this element.

The function `cf-synth` synthesizing the control flow links using the algorithm sketched above is explained in the next listings. It receives the sequence of statements `v`, the method's `Exit` node `exit`, the current loop's test expression (`loop-expr`), the statement following the current loop (`loop-succ`), and a map `label-succ-map` that assigns to each label reachable in the current scope the statement following the labeled statement. The `exit` parameter is used for handling return statements, and the last three parameters are used for handling break and continue statements. Initially, the function is called with `v` only containing the method, and `exit` bound to that method's `exit`. All other parameters are `nil`.

```
(defn cf-synth [v exit loop-expr loop-succ label-succ-map]
  (when (seq v)
    (let [[e1 & [n & _ :as tail]] v]
      (type-case e1
```

If the sequence `v` is not empty, its first element is bound to `e1`, and its rest is bound to `tail`. Furthermore, the first element of the rest (i.e., the second element of the sequence) is bound to `n`.

After binding these elements, a `type-case` dispatches on `e1`'s metamodel type. For example, if the element is a method, a control flow link to that method's first flow instruction is created, and the function recurses with the method's statements.

```
'flowgraph.Method (let [stmts (econtents el)]
                      (eadd! el :cfNext (cf-peek (first stmts)))
                      (recur stmts exit nil nil nil))
```

If the current element is a label, the function recurses with that label's statement prepended to the tail of the sequence. A mapping from this label to its following statement is added to the `label-succ-map`. This statement's first flow instruction is where the control flow continues when breaking to this label.

```
'flowgraph.Label (recur (cons (eget el :stmt) tail) exit loop-expr loop-succ
                       (assoc label-succ-map el n))
```

If the current element is a break statement, two cases have to be distinguished. If the break is labeled, a control flow link is added to the first flow instruction of the statement following the label which can be looked up in the `label-succ-map`. If the break is not labeled, a control flow link is added to the first flow instruction in the statement following the surrounding loop which is bound to `loop-succ`.

In any case, the function recurses with the tail of the sequence keeping all other parameters as-is.

```
'flowgraph.Break (do (if-let [l (eget el :label)]
                     (eadd! el :cfNext (cf-peek (label-succ-map l)))
                     (eadd! el :cfNext (cf-peek loop-succ)))
                  (recur tail exit loop-expr loop-succ label-succ-map))
```

There are similar cases for handling objects of the other metamodel types. The complete control flow transformation consists of 57 lines of code and is printed in Appendix C.

Task 3: Data Flow Analysis. The purpose of this task is to create `dfNext` links between `FlowInst` elements where the target element is a control flow successor of the source element, the target element uses (reads) a variable that was defined (written) by the source element, and the variable hasn't been rewritten in between. This definition has been implemented exactly as stated here, because although it's not the most efficient algorithm for the task, it is very clear and concise.

The function `find-nearest-definers` receives a flow instruction `fi` and a variable `uv` used by it, and it returns a vector of the nearest control flow predecessors that define that variable.

```
(defn find-nearest-definers [fi uv]
  (loop [preds (mapcat #(adjs % :cfPrev) (if (coll? fi) fi [fi])),
        r [], known #{}]
    (if (seq preds)
      (let [definers (filter #(member? uv (eget % :def)) preds)
            others (remove #(member? uv (eget % :def)) preds)]
        (recur (remove #(member? % known) (mapcat #(adjs % :cfPrev) others))
               (into r definers) (into known preds)))
      r)))
```

In Clojure, `loop` and `recur` implement a local tail-recursion, that is, inside a `loop` a `recur` form recurses not to the surrounding function but to the surrounding `loop`. Initially, `preds` is bound to the immediate control flow predecessors of `fi`, the result variable `r` is bound to the empty vector, and `known` is bound to the empty set.

If there are no predecessors, the result `r` is returned (the else-branch of the `if`). If there are control flow predecessors, those are sorted into `definers`, i.e., flow instructions that define `uv`. The other predecessors are sorted into `others`, i.e., flow instructions that don't write `uv`.

Then it is recursed to the surrounding `loop`. `preds` is rebound to those control flow predecessors of `others` that aren't already known in order not to recurse infinitely in case of control flow cycles, the

result vector r is rebound to the current r value plus the new definers, and $known$ is rebound to the union of the current $known$ value and the current $preds$.

The main function of this task simply uses this function to find the nearest definers of all flow instructions and their used variables and creates $dfNext$ links.

The complete data flow transformation consists of 19 lines of code and is printed in Appendix D.

Task 4: Control and Data Flow Validation. The goal of task 4 is to enable offloading testing effort for the transformations solving tasks 1 to 3 to programmers knowing only Java by equipping them with some easy to use DSL. The next listing shows an example validation specification as provided by the FunnyQT solution.

```
(make-test test-fg-transform-test0 "models/Test0.java.xmi"
  #{"testMethod()" "int a = 1;" ;; expected cfNext links
    ;; more [cf-predecessor cf-successor] tuples
    ["return b * c;" "Exit"]}
  #{"int a = 1;" "int c = a + b;" ;; expected dfNext links
    ;; more [df-predecessor df-successor] tuples
    ["b = a - b;" "return b * c;"]})
```

The FunnyQT solution uses Clojure’s metaprogramming facilities to create an *internal validation DSL*. `make-test` is a *macro*. A macro is a function that will be called by the Clojure compiler at compile-time. It receives the unevaluated arguments given to it, that is, its parameters are bound to code. Clojure, like all Lisps, is *homoiconic*, meaning that code is represented using usual Clojure data structures, e.g., lists, vectors, symbols, literals, etc. Thus, the macro is able to transform the code provided to it using standard Clojure functions to some new bunch of code that takes its place. Here, `make-test` creates a unit test that loads the given XMI model and compares it against the expected control and data flow links.

The complete macro implementation and two complete validation specifications are printed in Appendix E.

3 Evaluation

In this section, the FunnyQT solution to the Flowgraphs case is evaluated according to the criteria listed in the case description [Hor13].

All four tasks have been solved, and the results of every task are *complete and correct*. The FunnyQT solution consists of 313 lines of code excluding comments and empty lines, making it the shortest of all provided solutions. It is also the solution with the best *performance* and has won the *best efficiency award* for this case. However, because FunnyQT is a Clojure API with a functional alignment, its *understandability* depends largely on a reader’s prior knowledge about Clojure and functional programming.

References

- [HJSW09] Florian Heidenreich, Jendrik Johannes, Mirko Seifert, and Christian Wende. JaMoPP: The Java Model Parser and Printer. Technical Report TUD-FI09-10, Technische Universität Dresden, Fakultät Informatik, 2009. <ftp://ftp.inf.tu-dresden.de/pub/berichte/tud09-10.pdf>.
- [Hor13] Tassilo Horn. The TTC 2013 Flowgraphs Case. <https://github.com/tsdh/ttc-2013-flowgraphs-case/blob/master/desc/ttc-2013-flowgraphs-case.pdf?raw=true>, 2013.

A The complete JaMoPP-to-Text Transformation

```

1  (declare-polyfn stmt2str [elem])
2
3  (defn reduce-str [els]
4    (reduce #(str %1 (stmt2str %2)) "" els))
5
6  (defpolyfn stmt2str 'members.ClassMethod [method]
7    (str (aget method :name) "()"))
8
9  (defpolyfn stmt2str 'types.PrimitiveType [~org.eclipse.emf.ecore.EObject pt]
10   (clojure.string/lower-case (.getName (.eClass pt))))
11
12  (defpolyfn stmt2str 'statements.LocalVariableStatement [lv]
13    (let [v (aget lv :variable)]
14      (str (stmt2str (aget v :typeReference)) " " (stmt2str v)
15        (when-let [iv (aget v :initialValue)]
16          (str " = " (stmt2str iv)))
17          ";"))))
18
19  (defpolyfn stmt2str 'references.IdentifierReference [ir]
20    (stmt2str (aget ir :target)))
21
22  (defpolyfn stmt2str 'variables.Variable [v]
23    (aget v :name))
24
25  (defpolyfn stmt2str 'expressions.MultiplicativeExpression [me]
26    (let [[c1 c2] (aget me :children)]
27      (str (stmt2str c1) " " (reduce-str (aget me :multiplicativeOperators))
28          " " (stmt2str c2))))
29
30  (defpolyfn stmt2str 'expressions.EqualityExpression [ee]
31    (let [[c1 c2] (aget ee :children)]
32      (str (stmt2str c1) " " (reduce-str (aget ee :equalityOperators))
33          " " (stmt2str c2))))
34
35  (defpolyfn stmt2str 'expressions.AdditiveExpression [ae]
36    (let [[c1 c2] (aget ae :children)]
37      (str (stmt2str c1) " " (reduce-str (aget ae :additiveOperators))
38          " " (stmt2str c2))))
39
40  (defpolyfn stmt2str 'expressions.UnaryExpression [ue]
41    (str (reduce-str (aget ue :operators))
42        (stmt2str (aget ue :child))))
43
44  (defpolyfn stmt2str 'expressions.AssignmentExpression [ae]
45    (str (stmt2str (aget ae :child)) " "
46        (stmt2str (aget ae :assignmentOperator)) " "
47        (stmt2str (aget ae :value))))
48
49  (defpolyfn stmt2str 'expressions.RelationExpression [re]
50    (let [[c1 c2] (aget re :children)]
51      (str (stmt2str c1) " "
52          (reduce-str (aget re :relationOperators))
53          " " (stmt2str c2))))
54

```

```

55 (defpolyfn stmt2str 'expressions.SuffixUnaryModificationExpression [se]
56   (str (stmt2str (aget se :child))
57     (stmt2str (aget se :operator))))
58
59 (defpolyfn stmt2str 'statements.Block [b]
60   "\\{...\\}")
61
62 (defpolyfn stmt2str 'statements.Condition [c]
63   "if")
64
65 (defpolyfn stmt2str 'statements.WhileLoop [c]
66   "while")
67
68 (defpolyfn stmt2str 'statements.JumpLabel [l]
69   (str (aget l :name) ":"))
70
71 (defpolyfn stmt2str 'statements.Break [b]
72   (str "break"
73     (when-let [l (aget b :target)]
74       (str " " (aget l :name))))
75   ";"))
76
77 (defpolyfn stmt2str 'statements.Continue [c]
78   (str "continue"
79     (when-let [l (aget c :target)]
80       (str " " (aget l :name))))
81   ";"))
82
83 (defpolyfn stmt2str 'statements.Return [r]
84   (str "return" (when-let [rv (aget r :returnValue)]
85     (str " " (stmt2str rv))))
86   ";"))
87
88 (defpolyfn stmt2str 'statements.ExpressionStatement [stmt]
89   (str (stmt2str (aget stmt :expression)) ";"))
90
91 (defpolyfn stmt2str 'operators.Operator [op]
92   (type-case op
93     'operators.Multiplication "*"
94     'operators.Subtraction "-"
95     'operators.Addition "+"
96     'operators.Division "/"
97     'operators.LessThan "<"
98     'operators.GreaterThan ">"
99     'operators.Assignment "="
100    'operators.MinusMinus "--"
101    'operators.PlusPlus "++"
102    'operators.AssignmentPlus "+="
103    'operators.Equal "=="))
104
105 (defpolyfn stmt2str 'literals.Literal [l]
106   (type-case l
107     'literals.DecimalIntegerLiteral (aget l :decimalValue)))

```

B The complete JaMoPP-to-StructureGraph Transformation

```

1  (defn used-vars [s]
2    (reachables s [p-seq [p-* <>--]
3                      [p-restr 'references.IdentifierReference]
4                      :target]))
5
6  (deftransformation java2flowgraph [[in :emf] [out :emf]]
7    (^:top method2method [m]
8      :from 'members.ClassMethod
9      :to [fgm 'flowgraph.Method, ex 'flowgraph.Exit]
10     (eset! fgm :txt (stmt2str m))
11     (eset! ex :txt "Exit")
12     (eset! fgm :stmts (map stmt2item (seq (aget m :statements))))
13     (eset! fgm :exit ex)
14     (eset! fgm :def (map param2param (aget m :parameters))))
15    (stmt2item [stmt]
16      :generalizes [local-var-stmt2simple-stmt condition2if block2block
17                    return2return while-loop2loop break2break continue2continue
18                    label2label stmt2simple-stmt])
19    (var-creating-rule [v]
20      :generalizes [param2param local-var2var])
21    (param2param [p]
22      :from 'parameters.Parameter
23      :to [fgp 'flowgraph.Param]
24      (eset! fgp :txt (stmt2str p)))
25    (local-var2var [lv]
26      :from 'variables.LocalVariable
27      :to [fgv 'flowgraph.Var]
28      (eset! fgv :txt (stmt2str lv)))
29    (local-var-stmt2simple-stmt [lv]
30      :from 'statements.LocalVariableStatement
31      :to [fgss 'flowgraph.SimpleStmt]
32      (let [v (local-var2var (adj lv :variable))]
33        (eset! fgss :txt (stmt2str lv))
34        (eadd! fgss :def v)
35        (eset! fgss :use (map var-creating-rule
36                              (used-vars (adj lv :variable :initialValue))))))
37    (stmt2simple-stmt [s]
38      :from 'statements.Statement
39      :to [fgss 'flowgraph.SimpleStmt]
40      (eset! fgss :txt (stmt2str s))
41      (doseq [aex (reachables s [p-seq [p-* <>--]
42                                     [p-restr 'expressions.AssignmentExpression])]]
43        (eadd! fgss :def (var-creating-rule (the (used-vars (adj aex :child)))))
44        (eaddall! fgss :use (map var-creating-rule (used-vars (adj aex :value)))))
45      (doseq [umex (reachables s [p-seq [p-* <>--]
46                                     [p-restr 'expressions.UnaryModificationExpression])]]
47        (let [var (var-creating-rule (the (used-vars (adj umex :child)))]
48          (eadd! fgss :def var)
49          (eadd! fgss :use var))))
50    (label2label [l]
51      :from 'statements.JumpLabel
52      :to [fgl 'flowgraph.Label]
53      (eset! fgl :txt (stmt2str l))
54      (eset! fgl :stmt (stmt2item (aget l :statement))))

```



```

55 (expression2expr [ex]
56   :from 'expressions.Expression
57   :to [fgex 'flowgraph.Expr]
58   (eset! fgex :txt (stmt2str ex))
59   (eset! fgex :use (map var-creating-rule (used-vars ex))))
60 (condition2if [c]
61   :from 'statements.Condition
62   :to [fgif 'flowgraph.If]
63   (eset! fgif :txt (stmt2str c))
64   (eset! fgif :expr (expression2expr (eget c :condition)))
65   (eset! fgif :then (stmt2item (eget c :statement)))
66   (when-let [else (eget c :elseStatement)]
67     (eset! fgif :else (stmt2item else))))
68 (block2block [b]
69   :from 'statements.Block
70   :to [fgb 'flowgraph.Block]
71   (eset! fgb :txt (stmt2str b))
72   (eset! fgb :stmts (map stmt2item (eget b :statements))))
73 (return2return [r]
74   :from 'statements.Return
75   :to [fgr 'flowgraph.Return]
76   (eset! fgr :txt (stmt2str r))
77   (eset! fgr :use (map var-creating-rule (used-vars r))))
78 (break2break [b]
79   :from 'statements.Break
80   :to [fgb 'flowgraph.Break]
81   (eset! fgb :txt (stmt2str b))
82   (eset! fgb :label (label2label (eget b :target))))
83 (continue2continue [c]
84   :from 'statements.Continue
85   :to [fgc 'flowgraph.Continue]
86   (eset! fgc :txt (stmt2str c))
87   (eset! fgc :label (label2label (eget c :target))))
88 (while-loop2loop [wl]
89   :from 'statements.WhileLoop
90   :to [fgl 'flowgraph.Loop]
91   (eset! fgl :txt (stmt2str wl))
92   (eset! fgl :expr (expression2expr (eget wl :condition)))
93   (eset! fgl :body (stmt2item (eget wl :statement))))

```

C The complete Control Flow Transformation

```

1 (defn cf-peek [el]
2   (if (has-type? el 'flowgraph.FlowInstr)
3     el
4     (recur (first (econtents el)))))
5
6 (defn cf-synth [v exit loop-expr loop-succ label-succ-map]
7   (when (seq v)
8     (let [[el & [n & _ :as tail]] v]
9       (type-case el
10         'flowgraph.Method
11           (let [stmts (econtents el)]
12             (eadd! el :cfNext (cf-peek (first stmts)))
13             (recur stmts exit nil nil nil))

```

```

14   'flowgraph.SimpleStmt
15       (do (when n (eadd! el :cfNext (cf-peek n)))
16           (recur tail exit loop-expr loop-succ label-succ-map))
17   'flowgraph.Block
18       (recur (concat (econtents el) tail)
19           exit loop-expr loop-succ label-succ-map)
20   'flowgraph.Expr
21       (do (when n (eadd! el :cfNext (cf-peek n)))
22           (recur tail exit loop-expr loop-succ label-succ-map))
23   'flowgraph.Label
24       (recur (cons (eget el :stmt) tail) exit loop-expr loop-succ
25           (assoc label-succ-map el n))
26   'flowgraph.Return
27       (do (eadd! el :cfNext exit)
28           (recur tail exit loop-expr loop-succ label-succ-map))
29   'flowgraph.Break
30       (do (if-let [l (eget el :label)]
31           (eadd! el :cfNext (cf-peek (label-succ-map l)))
32           (eadd! el :cfNext (cf-peek loop-succ)))
33           (recur tail exit loop-expr loop-succ label-succ-map))
34   'flowgraph.Continue
35       (do (if-let [l (eget el :label)]
36           (eadd! el :cfNext (cf-peek l))
37           (eadd! el :cfNext loop-expr))
38           (recur tail exit loop-expr loop-succ label-succ-map))
39   'flowgraph.Loop
40       (let [[expr body] (econtents el)]
41           (recur (cons expr (cons body (cons expr tail)))
42               exit expr n label-succ-map))
43   'flowgraph.If
44       (let [[expr then else] (econtents el)]
45           (cf-synth [expr then (cf-peek n)]
46               exit loop-expr loop-succ label-succ-map)
47           (if else
48               (recur (cons expr (cons else tail))
49                   exit loop-expr loop-succ label-succ-map)
50               (recur (cons expr tail)
51                   exit loop-expr loop-succ label-succ-map)))
52   'flowgraph.Exit (assert (nil? n))))))
53
54 (defn synthesize-cf-edges [model]
55   (doseq [m (eallobjects model 'flowgraph.Method)]
56       :let [exit (the (eallobjects model 'flowgraph.Exit))]
57       (cf-synth [m] exit nil nil nil)))

```

D The complete Data Flow Transformation

```

1 (defn find-nearest-definers [fi uv]
2   (loop [preds (mapcat #(adjs % :cfPrev) (if (coll? fi) fi [fi]))
3         r []
4         known #{}]
5     (if (seq preds)
6         (let [definers (filter #(member? uv (eget % :def)) preds)
7               others (remove #(member? uv (eget % :def)) preds)]
8             (recur (remove #(member? % known) (mapcat #(adjs % :cfPrev) others))

```

```

9         (into r definers)
10        (into known preds)))
11    r)))
12
13 (defn synthesize-df-edges [model]
14   (doseq [fi (eallobjects model 'flowgraph.FlowInstr)
15         used-var (aget fi :use)
16         nearest-definer (find-nearest-definers fi used-var)]
17     (eadd! nearest-definer :dfNext fi))
18   (doseq [v (vec (eallobjects model 'Var))]
19       (edele! v)))

```

E The complete Validation DSL Implementation

```

1 (defn run-flowgraph-transformations [file]
2   (System/gc)
3   (println "Running Transformation on" file)
4   (print "Load Time: ")
5   (let [jamopp-model (time (load-model file))
6         outfile (str/replace (str/replace file "models/" "results/")
7                               ".java.xmi" ".xmi")
8         outvizfile (str/replace outfile ".xmi" ".pdf")
9         fg-trg (new-model)]
10    (println "Execution Times:")
11    (print " - JaMoPP to StructureGraph (with Vars): ")
12    (time (java2flowgraph jamopp-model fg-trg))
13    (print " - Control Flow Analysis: ")
14    (time (synthesize-cf-edges fg-trg))
15    (print " - Data Flow Analysis: ")
16    (time (synthesize-df-edges fg-trg))
17    (save-model fg-trg outfile)
18    (when (< (count (eallobjects fg-trg)) 80)
19        (print-model fg-trg outvizfile))
20    fg-trg))
21
22 (defmacro make-test [n file expected-cfs expected-dfs]
23   `(deftest ~n
24     (println "=====")
25     (let [fg-trg# (run-flowgraph-transformations ~file)
26           exp-cfs# ~expected-cfs
27           exp-dfs# ~expected-dfs
28           cfs# (set (map (fn [[s# t#]] [(aget s# :txt) (aget t# :txt)])
29                         (ecrosspairs fg-trg# :cfPrev :cfNext)))
30           dfs# (set (map (fn [[s# t#]] [(aget s# :txt) (aget t# :txt)])
31                         (ecrosspairs fg-trg# nil :dfNext)))]
32       (cond
33         (set? exp-cfs#) (let [cf-d1# (clojure.set/difference exp-cfs# cfs#)
34                             cf-d2# (clojure.set/difference cfs# exp-cfs#)]
35                           (is (empty? cf-d1#) "Missing cf-edges")
36                           (is (empty? cf-d2#) "Too many cf-edges"))
37         (number? exp-cfs#) (do
38                             (println "Only checking number of cfNext links.")
39                             (is (= exp-cfs# (count cfs#))))
40         :else (println "No expected cfNext links given.))
41       (cond

```

```

42 (set? exp-dfs#) (let [df-d1# (clojure.set/difference exp-dfs# dfs#)
43                       df-d2# (clojure.set/difference dfs# exp-dfs#)]
44   (is (empty? df-d1#) "Missing df-edges")
45   (is (empty? df-d2#) "Too many df-edges"))
46 (number? exp-dfs#) (do
47   (println "Only checking number of dfNext links.")
48   (is (= exp-dfs# (count dfs#))))
49 :else (println "No expected dfNext links given."))))

```

E.1 Two Example Validation Specifications

```

1 (make-test test-fg-transform-test4 "models/Test4.java.xmi"
2   #{"testMethod()" "int i = 100;"]
3   ["int i = 100;" "i > 0"]
4   ["i > 0" "Exit"]
5   ["i > 0" "i > 50"]
6   ["i > 50" "i--;"]
7   ["i > 50" "i = i - 10;"]
8   ["i = i - 10;" "i == 50"]
9   ["i == 50" "break;"]
10  ["i == 50" "i > 50"]
11  ["break;" "i--;"]
12  ["i--;" "i > 0"]}
13  #{"int i = 100;" "i > 0"]
14  ["int i = 100;" "i > 50"]
15  ["int i = 100;" "i = i - 10;"]
16  ["int i = 100;" "i--;"]
17  ["i = i - 10;" "i == 50"]
18  ["i = i - 10;" "i > 50"]
19  ["i = i - 10;" "i = i - 10;"]
20  ["i = i - 10;" "i--;"]
21  ["i--;" "i > 0"]
22  ["i--;" "i > 50"]
23  ["i--;" "i = i - 10;"]
24  ["i--;" "i--;"]})
25
26 ;; For the large models, only the correct number of cfNext/dfNext links is asserted.
27 (make-test test-fg-transform-test9 "models/Test9.java.xmi" 14452 27202)

```