# Solving the TTC 2013 Flowgraphs Case with FunnyQT

Dipl.-Inform. Tassilo Horn
horn@uni-koblenz.de
Institute for Software Technology, University Koblenz-Landau, Germany

This paper describes the FunnyQT solution to the TTC 2013 Flowgraphs Transformation Case.

FunnyQT is a model querying and model transformation library for the functional Lisp-dialect Clojure providing a rich and efficient querying and transformation API.

The FunnyQT solution of the Flowgraphs case solves all four tasks, and it has won the *best efficiency award* for this case.

## 1 Introduction

*FunnyQT* is a new model querying and transformation approach which is implemented as an API for the functional, JVM-based Lisp-dialect Clojure. It provides several sub-APIs for implementing different kinds of queries and transformations. For example, there is a model-to-model transformation API, and there is an in-place transformation API for writing programmed graph transformations.

For solving the tasks of this transformation case[1], FunnyQT's model transformation API and its polymorphic function API have been used for task 1. Both task 2 and task 3 have been tackled algorithmically using FunnyQT's plain querying and model manipulation APIs. Task 4 has been solved by using FunnyQT's querying API and Clojure metaprogramming.

## 2 Solution Description

**Task 1: JaMoPP to StructureGraph.** According to the case description [Hor13], the goal of this task is to transform a fine-granular Java syntax graph conforming to the JaMoPP metamodel [HJSW09] into a much simpler structure graph model that only contains statements and expressions that are neither structured nor subdivided any further. However, the original Java code of these statements and expressions should be reflected in the new elements' `txt` attribute. This model-to-text transformation is described in Section 2. Thereafter, the model-to-model transformation creating a structure graph from a JaMoPP model is described in Section 2.

**JaMoPP to Text.** This model-to-text transformation is implemented using FunnyQT's polymorphic function API. A polymorphic function is a function that is declared once, and then arbitrary many implementations for concrete metamodel types can be added. When a polymorphic function is called, the actual implementation is determined similarly to the typical dispatch in object-oriented programming languages. If there's no implementation provided for the element's type or one of its supertypes, an exception is thrown.

The function `stmt2str` implements the model-to-text transformation required for solving task 1. It is declared as follows.

[1] This FunnyQT solution is available at `https://github.com/tsdh/ttc-2013-flowgraphs` and on SHARE (image TTC13::Ubuntu12LTS_TTC13::FunnyQT.vdi

Submitted to:
Transformation Tool Contest 2013 (TTC'13)

```
(declare-polyfn stmt2str [elem])
```

`declare-polyfn` declares a new polymorphic function. Its name is `stmt2str`, and it receives exactly one parameter `elem`. It's task is to create a string representation matching the concrete Java syntax for the provided JaMoPP model element.

After the polymorphic function has been declared, implementations for concrete metamodel types can be added using `defpolyfn`. For example, this is the implementation for JaMoPP elements of type `AssignmentExpression`:

```
(defpolyfn stmt2str 'expressions.AssignmentExpression [ae]
  (str (stmt2str (eget ae :child)) " "
       (stmt2str (eget ae :assignmentOperator)) " "
       (stmt2str (eget ae :value))))
```

The `child` of the assignment expression is some variable, the assignment operator is one of `=`, `+=`, `-=`, `*=`, or `/=`, and `value` is an arbitrary expression. These three components are converted to strings recursively which are then concatenated.

All in all, the polymorphic `stmt2str` function consists of 22 implementations for various JaMoPP metamodel types accounting to a total of about 120 lines of code.

**JaMoPP to Structure Graph.** The model-to-model transformation part of task 1 is implemented using FunnyQT's model-to-model transformation API. This transformation also creates `Var` and `Param` objects as requested by task 3.1.

The transformation starts by defining its name and input and output models.

```
(deftransformation java2flowgraph [[in :emf] [out :emf]]
```

There could be arbitrary many input and output models, and they could be of different kinds, e.g., a transformation could receive a JGraLab TGraph and some EMF model, and create an output EMF model. Here, it gets only the JaMoPP EMF input model which is bound the the variable `in`, and one single structure graph output model bound to `out`, which is also an EMF model.

In the body of such a transformation, arbitrary many rules may be declared. The first rule is the `method2method` rule shown in the next listing. The `^:top` metadata preceeding the rule name specifies that the rule is a top-level rule. Such rules are applied to all matching elements by the transformation itself, whereas non-top-level rules have to be called explicitly from a top-level rule (directly or indirectly).

```
  (^:top method2method [m]
         :from 'members.ClassMethod
         :to [fgm 'flowgraph.Method, ex 'flowgraph.Exit]
         (eset! fgm :txt (stmt2str m)) ;; Invoke the model-to-text transformation
         (eset! ex :txt "Exit")
         (eset! fgm :exit ex)
         (eset! fgm :stmts (map stmt2item (eget m :statements)))
         (eset! fgm :def (map param2param (eget m :parameters))))
```

It receives a model element `m`. The `:from` clause dictates that `m` must be of type `ClassMethod` in order for the rule to be applicable. The `:to` clause declares the objects to be created. Here, for a given JaMoPP method, a corresponding flowgraph method and its exit object are created. The remainder of the rule is its body. Here, the `txt` attribute of the new method and its exit are set, the former using the polymorphic `stmt2str` function discussed in Section 2. The method's `stmts` reference is set by applying another rule, `stmt2item`, to the statements of the JaMoPP method. Likewise, the method's parameters are transformed by mapping them to the `param2param` rule for setting the method's `def` reference.

A special kind of rules are generalizing rules such as the one shown in the next listing.

```
(stmt2item [stmt]
    :generalizes [local-var-stmt2simple-stmt condition2if block2block
                  return2return while-loop2loop break2break continue2continue
                  label2label stmt2simple-stmt])
```

This concept is quite similar to mapping disjunction in QVT Operational Mappings. When this rule is called, the rules specified in the `:generalizes` vector are tried one after the other, and the first applicable one is applied, and its result is returned.

The complete `java2flowgraph` model-to-model transformation consists of 15 rules with 94 lines of code in total.

**Task 2: Control Flow Analysis.** The purpose of this task is to create `cfNext` links between `FlowInstr` elements in the flowgraph model created by the model-to-model transformation realizing task 1. Every such flow instruction should be connected to every other flow instruction that may be the next one in the program's control flow. This challenge has been tackled algorithmically using FunnyQT's plain quering and model manipulation APIs.

The algorithm uses a sequence of statements as intermediate representation to work on realizing a pre-order depth-first traverlal with look-ahead through the method's statements. In the general case, every flow instruction in that sequence has to be connected with the immediately following flow instruction in the sequence. For various kinds of statements, special rules are needed. For example, when encountering a block in the sequence (which is no flow instruction), the the block is replaced with its contents.

Since the next statement in the sequence might not be a flow instruction but some structured statement like a block, an if-statement, or a loop, there's a helper function `cf-peek`. It receives some element and returns either this element if it is a flow instruction, or otherwise the first flow instruction inside this element.

The function `cf-synth` synthesizing the control flow links using the algorithm sketched above is explained in the next listings. It receives the sequence of statements `v`, the method's `Exit` node `exit`, the current loop's `loop-expr`, the statement following the current loop `loop-succ`, and a map `label-succ-map` that assigns to each label reachable in the current scope the statement following the labeled statement. The `exit` parameter is used for handling return statements, and the last three parameters are used for handling break and continue statements. Initially, the function is called with `v` only containing the method, and `exit` bound to that method's exit. All other parameters are `nil`.

```
(defn cf-synth [v exit loop-expr loop-succ label-succ-map]
  (when (seq v)
    (let [[el & [n & _ :as tail]] v]
      (type-case el
```

If the sequence `v` is not empty, its first element is bound to `el`, and its rest is bound to `tail`. Furthermore, the first element of the rest (i.e., the second element of the sequence) is bound to `n`.

After binding these elements, a `type-case` dispatches on `el`'s metamodel type. For example, if the element is a method, a control flow link to that method's first flow instruction is created, and the function recurses with the method's statements.

```
        'flowgraph.Method (let [stmts (econtents el)]
                            (eadd! el :cfNext (cf-peek (first stmts)))
                            (recur stmts exit nil nil nil))
```

If the current element is a label, the function recurses with that label's statement prepended to the tail of the sequence. A mapping from this label to its following statement is added to the `label-succ-map`. This statement's first flow instruction is where the control flow continues when breaking to this label.

```
'flowgraph.Label (recur (cons (eget el :stmt) tail) exit loop-expr loop-succ
                        (assoc label-succ-map el n))
```

If the current element is a break statement, two cases have to be distinguished. If the break is labled, a control flow link is added to the first flow instruction of the statement following the label which can be looked up in the `label-succ-map`.

If the break is not labled, a control flow link is added to the first flow instruction in the statement following the surrounding loop which is bound to `loop-succ`.

In any case, the function recurses with the tail of the sequence keeping all other parameters as-is.

```
'flowgraph.Break (do (if-let [l (eget el :label)]
                       (eadd! el :cfNext (cf-peek (label-succ-map l)))
                       (eadd! el :cfNext (cf-peek loop-succ)))
                   (recur tail exit loop-expr loop-succ label-succ-map))
```

There are similar cases for handling objects of the other metamodel types. The complete control flow transformation consists of 55 lines of code.

**Task 3: Data Flow Analysis.** The purpose of this task is to create `dfNext` links between `FlowInst` elements where the target element is a control flow successor of the source element, the target element uses (reads) a variable that was defined (written) by the source element, and the variable hasn't been rewritten in between. This definition has been implemented exactly as stated here, because although it's not the most efficient algorithm for the task, it is very clear and concise.

The function `find-nearest-definers` receives a flow instruction `fi` and a variable `uv` used by it, and it returns a vector of the nearest control flow predecessors that define that variable.

```
(defn find-nearest-definers [fi uv]
  (loop [preds (mapcat #(adjs % :cfPrev) (if (coll? fi) fi [fi])),
         r [], known #{}]
    (if (seq preds)
      (let [definers (filter #(member? uv (eget % :def)) preds)
            others   (remove #(member? uv (eget % :def)) preds)]
        (recur (remove #(member? % known) (mapcat #(adjs % :cfPrev) others))
               (into r definers) (into known preds)))
      r)))
```

In Clojure, `loop` and `recur` implement a local tail-recursion, that is, inside a `loop` a `recur` form recurses not to the surrounding function but to the surrounding `loop`. Initially, `preds` is bound to the immediate control flow predecessors of `fi`, the result variable `r` is bound to the empty vector, and `known` is bound to the empty set.

If there are no predecessors, the result `r` is returned (the else-branch of the if). If there are control flow predecessors, those are sorted into `definers`, i.e., flow instructions that define `uv`. The other predecessors are sorted into `others`, i.e., flow instructions that don't write `uv`.

Then it is recursed to the surrounding `loop`. `preds` is rebound to those control flow predecessors of `others` that aren't already known in order not to recurse infinitely in case of control flow cycles, the result vector `r` is rebound to the current `r` value plus the new `definers`, and `known` is rebound to the union of the current `known` value and the current `preds`.

The complete data flow transformation consists of 18 lines of code.

**Task 4: Control and Data Flow Validation.** The goal of task 4 is to enable offloading testing effort for the transformations solving tasks 1 to 3 to programmers knowing only Java by equipping them with some easy to use DSL. The next listing shows an example validation specification as provided by the FunnyQT solution.

```
(make-test test-fg-transform-test0 "models/Test0.java.xmi"
           #{["testMethod()"   "int a = 1;"]     ;; expected cfNext links
             ;;...
             ["return b * c;"  "Exit"]}
           #{["int a = 1;"     "int c = a + b;"] ;; expected dfNext links
             ;;...
             ["b = a - b;"     "return b * c;"]})
```

The FunnyQT solution uses Clojure's metaprogramming facilities to create an *internal validation DSL*. `make-test` is a *macro*. A macro is a function that will be called by the Clojure compiler at compile-time. It receives the unevaluated arguments given to it, that is, its parameters are bound to code. Clojure, like all Lisps, is *homoiconic*, meaning that code is represented using usual Clojure data structures, e.g., lists, vectors, symbols, literals, etc. Thus, the macro is able to transform the code provided to it using standard Clojure functions to some new bunch of code that takes its place. Here, `make-test` will create a unit test that loads the provided model and compares it against the expected control and data flow links.

## 3 Evaluation

In this section, the FunnyQT solution to the Flowgraphs case is evaluated according to the critera listed in the case description [Hor13].

All four tasks have been solved, and the results of every task are complete and correct. The FunnyQT solution consists of 335 lines of code excluding comments and empty lines, making it the shortest of all provided solutions. It is also the solution with the best performance and has won the *best efficiency award* for this case.

## References

[HJSW09] Florian Heidenreich, Jendrik Johannes, Mirko Seifert, and Christian Wende. JaMoPP: The Java Model Parser and Printer. Technical Report TUD-FI09-10, Technische Universität Dresden, Fakultät Informatik, 2009. `ftp://ftp.inf.tu-dresden.de/pub/berichte/tud09-10.pdf`.

[Hor13] Tassilo Horn. The TTC 2013 Flowgraphs Case. `https://github.com/tsdh/ttc-2013-flowgraphs-case/blob/master/desc/ttc-2013-flowgraphs-case.pdf?raw=true`, 2013.